The Lanczos tensor is an interesting animal. It can be thought of as the source of the Weyl curvature tensor, the traceless part of the Riemann curvature tensor.

The Weyl tensor and the Ricci tensor together fully determine the Riemann tensor, i.e., the intrinsic curvature of a spacetime. Crudely put, whereas the Ricci tensor tells you how the volume of, say, a cloud of dust changes in response to gravity, the Weyl tensor tells you how that cloud of dust is distorted in response to the same gravitational field. (For instance, consider a cloud of dust in empty space falling towards the Earth. In empty space, the Ricci tensor is zero, so the volume of the cloud does not change. But its shape becomes distorted and elongated in response to tidal forces. This is described by the Weyl tensor.)

Because the Ricci tensor is absent, the Weyl tensor fully describes gravitational fields in empty space. In a sense, the Weyl tensor is analogous to the electromagnetic field tensor that fully describes electromagnetic fields in empty space. The electromagnetic field tensor is sourced by the four-dimensional electromagnetic vector potential (meaning that the electromagnetic field tensor can be expressed using partial derivatives of the electromagnetic vector potential.) The Weyl tensor has a source in exactly the same sense, in the form of the Lanczos tensor.

The electromagnetic field does not uniquely determine the electromagnetic vector potential. This is basically how integrals vs. derivatives work. For instance, the derivative of the function $y=x^2$ is given by $y'=2x$. But the inverse operation is not unambiguous: $\int 2x~ dx=x^2+C$ where $C$ is an arbitrary integration constant. This is a recognition of the fact that the derivative of any function in the form $y=x^2+C$ is $y'=2x$ regardless of the value of $C$; so knowing only the derivative $y'$ does not fully determine the original function $y$.

In the case of electromagnetism, this freedom to choose the electromagnetic vector potential is referred to as the gauge freedom. (And this gauge freedom can be used to deduce the electromotive force, but that is another story.) A similar gauge freedom also exists for the Lanczos tensor. And this gauge freedom is what led to an interesting question during a recent e-mail conversation I had with someone: could the Lanczos tensor be used in some way by an observer to determine of he is at or near the event horizon of a black hole? Conventional wisdom says no, since for a freely falling observer, the event horizon is not in any way special. But could conventional wisdom be wrong?

The Weyl-Lanczos equation is given by

\begin{align}C_{abcd}&=H_{abc;d}+H_{cda;b}+H_{bad;c}+H_{dcb;a} \\ &\, \, \, \, \, + (H^e{}_{(ac);e} + H_{(a|e|}{}^e{}_{;c)})g_{bd} + (H^e{}_{(bd);e} + H_{(b|e|}{}^e{}_{;d)})g_{ac} \\ &\, \, \, \, \, - (H^e{}_{(ad);e} + H_{(a|e|}{}^e{}_{;d)})g_{bc} - (H^e{}_{(bc);e} + H_{(b|e|}{}^e{}_{;c)})g_{ad} \\ &\, \, \, \, \, -\frac{2}{3} H^{ef}{}_{f;e}(g_{ac}g_{bd}-g_{ad}g_{bc}),\end{align}

where $C_{abcd}$ is the Weyl tensor, $H_{abc}$ is the Lanczos tensor, and $g_{ab}$ is the metric. Covariant derivatives with respect to the metric are indicated by the semicolon, while round brackets in indices indicate symmetrization.

For the Schwarzschild metric $ds^2=(1-2m/r)dt^2-(1-2m/r)^{-1}dr^2-r^2d\theta^2-r^2\sin^2\theta d\phi^2$, the simplest solution is given by

\begin{align}H_{trt}=-H_{rtt}=\frac{m}{r^2},\end{align}

where $m$ is the source mass, $r$ is the radial Schwazschild coordinate and we choose units such that the universal gravitational constant and the speed of light are both 1. All other components of the Lanczos tensor are zero.

However, because of the gauge freedom, other solutions exist. A notable solution is the solution given in the "Lanczos algebraic gauge", a gauge choice that greatly simplifies the Weyl-Lanczos equation. This solution is given by

\begin{align}H_{trt}&=-H_{rtt}=\frac{2m}{3r^2},\\
H_{r\theta\theta}&=-H_{\theta r\theta}=-\frac{m}{3(1-2m/r)},\\
H_{r\phi\phi}&=-H_{\phi r\phi}=-\frac{m~\sin^2\theta}{3(1-2m/r)}.\end{align}

There is one property shared by these two solutions. The invariant scalar quantity $H_{abc}H^{abc}$ in both cases is infinite at the horizon where $r=2m$ (the value is $-2m^2/r^3(r-2m)$ and $-4m^2/3r^3(r-2m)$, respectively, and it is the $r-2m$ bit in the denominator that causes trouble.) So the question is, then, would this be true for all solutions of the Weyl-Lanczos equation in the Schwarzschild metric?

The answer is no. The gauge freedom of the Lanczos tensor is expressed by the fact that if $H_{abc}$ is a solution of the Weyl-Lanczos equation, then so is

$$H'_{abc} = H_{abc} + \Phi_{[a}g_{b]c},$$

where $\Phi_a$ is an arbitrary vector field and the square brackets indicate antisymmetrization.

So if we start with the solution given by $H_{trt}=-H_{rtt}=m/r^2$, the condition for $H_{abc}H^{abc}$ to be zero at the horizon is given by

$$\Phi_t^2 = -\frac{2m(r-2m)}{3r^3}\Phi_r + \frac{(r-2m)^2}{r^2}\Phi_r^2 + \frac{r-2m}{r^3}\Phi_\theta^2+\frac{r-2m}{r^3\sin^2\theta}\Phi_\phi^2+\frac{m^2}{3r^4}.$$

In particular, the solution given by $\Phi_t = m/\sqrt{3}r^2$ with all other components of $\Phi$ being 0 is such a solution. The nontrivial components of this solution are:

\begin{align}H_{trt}&= \frac{m}{r^2},\\
H_{trr}&= \frac{m}{\sqrt{3}r(2m-r)},\\
H_{t\theta\theta}&= -\frac{m}{\sqrt{3}},\\
H_{t\phi\phi}&= -\frac{m~\sin^2\theta}{\sqrt{3}}.\end{align}

This can be verified by direct substitution as a valid solution of the Weyl-Lanczos equation (and furthermore, the above equation for $\Phi_t$ provides an infinite 3-parameter set of such solutions) for which $H_{abc}H^{abc} = 0$ everywhere.

I used the following Maxima code to study the Lanczos tensor:

load(ctensor);
ct_coordsys(exteriorschwarzschild);
lg:-lg;
cmetric();
christof(false);
lriemann(false);
riemann(false);
weyl(false);
imetric(g);
components(H([a,m],[]),g([],[r,s])*(covdiff(H([a,s,m],[]),r)-covdiff(H([a,s,r],[]),m)));
components(g([a,b,c,d],[]),g([a,c],[])*g([b,d],[])-g([a,d],[])*g([b,c],[]));
components(C([b,c,d,a],[]),
(covdiff(H([a,b,c],[]),d)-covdiff(H([a,b,d],[]),c)+
covdiff(H([c,d,a],[]),b)-covdiff(H([c,d,b],[]),a))+
1/2*(H([a,d],[])*g([b,c],[])+H([d,a],[])*g([b,c],[])+
H([b,c],[])*g([a,d],[])+H([c,b],[])*g([a,d],[])-
H([a,c],[])*g([b,d],[])-H([c,a],[])*g([b,d],[])-
H([b,d],[])*g([a,c],[])-H([d,b],[])*g([a,c],[]) )+
2/3*g([],[e,i])*g([],[f,j])*covdiff(H([i,j,e],[]),f)*g([a,b,c,d],[])
);

EQ:W([a,b,c,d],[])=C([a,b,c,d],[])$SOL:ic_convert(EQ)$

for i thru 4 do for j thru 4 do for k thru 4 do HH[i,j,k]:0;
HH[1,2,1]:m/r^2;
HH[2,1,1]:-m/r^2;

EHH:ic_convert(H([a,b,c],[])=HH([a,b,c],[])+F([a],[])*g([b,c],[])-F([b],[])*g([a,c],[]));

ev(EHH);

for i thru dim do for j thru dim do for k thru dim do if H[i,j,k]#0 then ldisplay(H[i,j,k]);

ev(SOL);

for i thru 4 do for j thru 4 do for k thru 4 do for l thru 4 do weyl[i,j,k,l]:factor(weyl[i,j,k,l]);for i thru 4 do for j thru 4 do for k thru 4 do for l thru 4 do W[i,j,k,l]:factor(W[i,j,k,l]);for i thru dim do for j from i+1 thru dim do for k from i thru dim do for l thru dim do if W[i,j,k,l] # 0 then ldisplay(W[i,j,k,l]);for i thru dim do for j from i+1 thru dim do for k from i thru dim do for l thru dim do if weyl[i,j,k,l] # 0 then ldisplay(weyl[i,j,k,l]);for i thru 4 do for j thru 4 do for k thru 4 do for l thru 4 do Z[i,j,k,l]:factor(W[i,j,k,l]-weyl[i,j,k,l]);for i thru 4 do for j thru 4 do for k thru 4 do for l thru 4 do if Z[i,j,k,l] # 0 then ldisplay(Z[i,j,k,l]);H0:0;
for i thru 4 do for j thru 4 do for k thru 4 do H0:factor(H0+H[i,j,k]*H[i,j,k]*ug[i,i]*ug[j,j]*ug[k,k]);
H0:factor(H0);

solve(H0=0,F[1]^2);
factor(H0),%[1];